

13

שיעור

מערכים

מהו מערך?

מערך הוא מעין משתנה, המכיל מספר איברים, תחת שם אחד, כאשר לכל איבר במערך יש מספר אינדקס המזהה אותו.

הצהרה על מערך

ההצהרה על מערך דומה מאוד להצהרה על משתנה, לדוגמא:

```
Dim Products(9) As String
```

בשורה זו הצהרנו על מערך בשם Products ובו 10 איברים, שכל אחד מהאיברים הוא משתנה מסוג מחרוזת (String).
למה במערך שלמעלה יש 10 איברים, הרי הצהרנו רק על 9?
התשובה לכך היא, שויזואל בייסיק, כמו רוב שפות התכנות, מתחילות לספור מ-0 ולא מ-1 ולכן מ-0 עד 9 ישנם 10 איברים. מאוחר יותר בשיעור נראה איך אפשר לגרום לויזואל בייסיק להתחיל לספור מ-1 ולא מ-0.

הצבת ערכים לאיברים במערך

אחרי שהצהרנו על המערך, ניתן להציב ערכים לאיבריו, למשל:

```
Products(2) = "Keyboard"
```

בשורה שלמעלה, הצבנו לאיבר, בעל האינדקס 2 (שהוא האיבר השלישי במערך, כי ויזואל בייסיק מתחילה לספור מ-0), את המחרוזת "Keyboard". בדומה ניתן להציב גם לאיברים אחרים מחרוזות שונות, למשל:

```
Products(9) = "Motherboard"
```

אם ננסה להציב מחרוזת לאיבר, שהאינדקס שלו 10, נקבל את השגיאה הבאה:
Subscript out of range, כלומר חרגנו מהתחום של המערך כי במערך קיימים האיברים 0 עד 9 בלבד, ואיבר 10 אינו קיים.

שימוש בערכים ממערך

ניתן כמובן לשמור ערכים במערך, ואחר כך להציב אותם למשתנים שונים, למשל:
פתח פרוייקט חדש, הוסיפו לפרוייקט תיבת טקסט אחת (והשאירו את שמה Text1) וכפתור פקודה אחד (ותשאירו את שמו Command1). כעת הוסיפו בעורך הקוד את הקוד הבא:

```
1 Private Sub Command1_Click()
2   Dim Prices(2) As Integer
3   Prices(0) = 100
4   Prices(1) = 200
5   Prices(2) = 250
6   Text1.Text = Prices(1)
7 End Sub
```

בשורה 1 מתחיל הנוהל במקרה שהמשתמש ילחץ על הכפתור Command1.
 בשורה 2 הצהרנו על מערך בעל 3 איברים, שכל איבר הוא מסוג מספר שלם (Integer).
 בשורות 3-5 הצבנו לשלושת האיברים שבמערך, ערכים שונים.
 בשורה 4, אנו משנים את הטקסט של תיבת הטקסט לערך השמור באיבר 1 (בעל האינדקס 1 ולא הראשון), כלומר הטקסט בתיבת הטקסט יראה כעת את הערך 200.

הגדרת טווח האינדקסים במערך

אם הנכם רוצים שאינדקס של האיבר הראשון במערך יהיה 1 (ולא 0) ושל האחרון יהיה 10, ניתן להצהיר על המערך בצורה הבאה:

Dim Prices(1 To 10) As Integer

כלומר הצהרנו על מערך בן 10 איברים, שהאיבר הראשון הוא 1 והאיבר האחרון הוא 10. ניתן גם שהמערך יתחיל מאיזה אינדקס שתמצאו למשל:

Dim Parts(55 To 99) As Long

בשורה למעלה, הצהרנו על מערך שהאיבר הראשון בו, הוא בעל האינדקס 55 והאחרון הוא בעל האינדקס 99.

בנוסף ניתן לקבוע באופן ידני בוויזואל בייסיק, האם האינדקס של האיבר הראשון יהיה 0 או 1, על ידי הוספת ההצהרה Option Base 1 לשורה הראשונה בעורך הקוד:

Option Base 1

ההצהרה הבאה אומרת, שכאשר לא מצהירים על תחום של האינדקסים במערך, האיבר הראשון במערך יהיה 1 ולא 0, לדוגמא:

Option Base 1

Private Sites(10) As String

במקרה הזה האינדקס של האיבר הראשון במערך ששמו Sites יהיה 1 ולא 0. אם רוצים שהאינדקס של האיבר יהיה 0, או שמצהירים בראש עורך הקוד את ההצהרה הבאה:

Option Base 0

או שלא מצהירים כלום, ואוטומטית האינדקס של האיבר הראשון יהיה 0. אם רוצים שהאינדקס הראשון יהיה איבר אחר יש להשתמש בהצהרה עם To, כמו שראינו קודם, למשל:

Private Sites(20 To 30) As String

טווח ההצהרה של מערך

הכללים לגבי טווח ההצהרה של המערך זהים לחלוטין לכללי טווח ההצהרה של משתנים. לפרטים נוספים על טווח ההצהרה פנו לשיעור על משתנים (שיעור 10).

מערך דינמי

לפעמים אנו לא יודעים כמה איברים אנו צריכים במערך. למשל שמות הספרים שיש בחנות ספרים, פתרון אחד הוא להצהיר על מספר פריטים רב במערך, למשל:

Dim Books(999) As String

הצהרנו כאן על מערך בשם Books ובו 1000 איברים מסוג מחרוזת (String). אך יש לכך מספר חסרונות, למשל אם מספר הספרים הוא מועט, נגיד 20, אנו מבזבזים זיכרון רב שאינו מנוצל. חסרון שני הוא, נגיד שקנינו כמות גדולה של ספרים לחנות שעולה על 1000 ספרים, אז לא נוכל לאכסן את שמות הספרים מעבר ל-1000 ספרים. הפתרון לכך הוא מערך דינמי. מערך דינמי הוא מערך שלא מגדירים בו את מספר האיברים מראש, ואחר כך אם מספר האיברים גדל או קטן במערך, ניתן לשנותו בהתאם. ההצהרה על מערך דינמי, נראית כך:

Dim Books() As String

ניתן לראות כי ההצהרה דומה מאוד, להצהרה על מערך רגיל, מלבד העובדה שהסוגריים ריקים, כלומר שלא הצהרנו מראש על מספר האיברים במערך.
עכשיו נגיד שאנו רוצים לאחסן 5 פריטים במערך, אז נשנה את גודלו של המערך באמצעות ההצהרה ReDim, למשל:

ReDim Books(4) As String

וכעת ניתן להציב חמישה איברים במערך, מהאינדקס 0 עד האינדקס 4.
וניתן להקטין בחזרה את המערך, שנית באמצעות ReDim, למשל:

ReDim Books(1) As String

כאן הקטנו את מספר האיברים במערך ל-2. אך המינימום להקטנת מערך באמצעות ReDim הוא לאיבר אחד, למשל כך:

ReDim Books(0) As String

שמירה על הנתונים במערך דינמי

אך ישנה בעיה חמורה עם שינוי גודלו של מערך דינמי, למשל אם הצהרנו על מערך דינמי שינינו את גודלו, והצבנו נתונים לאיברים במערך. בפעם הבאה שנשנה את גודל המערך האיברים הללו ימחקו ונצטרך להציב מחדש ערכים לאיברים הללו.

אך ניתן גם לשנות את גודלו של המערך הדינמי מבלי לאבד את הנתונים, על ידי מילת המפתח Preserve (לשמור), למשל:

1	Dim Movies() As String
2	ReDim Movies (0) As String
3	Movies (0) = " Gone with the wind"
4	ReDim Preserve Movies (19) As String

בשורה 1, הצהרנו על מערך דינמי בשם Books.
בשורה 2, שינינו את גודל המערך לאיבר אחד.
בשורה 3, הצבנו לאיבר בעל האינדקס 0, את המחרוזת "Gone with the wind".
בשורה 4, הגדלנו את המערך ל-20 איברים, תוך כדי שימור האיברים הקודמים, כלומר הערך של איבר 0 נשאר המחרוזת "Gone with the wind". אם לא היינו משתמשים במילת המפתח Preserve המחרוזות שמאוחסנת באיבר 0, היתה נמחקת.

מחיקת הערכים מכל האיברים במערך – הפקודה Erase

לפעמים אנו רוצים למחוק את הערכים מכל האיברים במערך, למשל הצהרנו על המערך הבא:

Dim Brands(10) As String

הצבנו למספר איברים שלו ערכים:

Brands(0) = "Microsoft"
Brands(1) = "Borland"
Brands(2) = "CA"

וכעת אנו רוצים למחוק את התוכן של כל האיברים.
נעשה זאת באמצעות הפקודה Erase, לדוגמא:

Erase Brands

בשורה זאת אפסנו את התוכן של כל אחד מהאיברים במערך.
במערך שאיבריו הם מחרוזות (String) הערך לאחר המחיקה יהיה מחרוזת ריקה, במערך שאיבריו הם מסוג מספרי (כמו Integer או Long), הערך לאחר המחיקה יהיה 0, במערך שאיבריו מסוג Variant, הערך לאחר המחיקה יהיה Empty (משתנה ריק), ובמקרה של אובייקטים הערך יהיה Nothing.
למשל כעת ערכו של האיבר Brands(1) הוא מחרוזת ריקה, כלומר "".

מערך רב מימדי

למערכים בהם השתמשנו עד כה קוראים גם "מערכים שטוחים" (flat arrays), כי יש להם רק מימד אחד. אבל לעיתים העניינים יותר מורכבים, למשל נגיד שיש לנו טבלה שיש בה תאים רבים (מעין מטריצה), ואנו רוצים לזהות כל תא גם באמצעות השורה והעמודה בו הוא נמצא, למשל, יש לנו את הטבלה הבאה:

	1	2	3	4	5	6	7	8
1		ADA					Eiffel	
2	Assembler			Delphi		Logo		C#
3		Pascal			VB			
4			BASIC					Cobol
5	C++			C	Prolog		Fortran	
6		Java				Perl		

טבלה זו מורכבת מ-6 שורות ו-8 טורים.

נצוהיר על מערך רב מימדי המייצג את הטבלה הזו, כך:

`Dim Table(1 To 6, 1 To 8) As String`

כלומר הצהרנו על מערך בשם Table בעל שני מימדים, שבמימד הראשון (המייצג במקרה שלנו את השורות שבטבלה) ישנם 6 איברים, ובמימד השני (המייצג במקרה שלנו את הטורים שבטבלה) ישנם 8 איברים.

כעת אם נרצה לאכסן את הערך "VB", הנמצא בשורה השלישית ובטור החמישי, נעשה זאת כך:

`Table(3, 5) = "VB"`

בדומה אם נרצה לאכסן את הערך "Fortran" הנמצא בשורה החמישית ובטור השביעי, נעשה זאת כך:

`Table(5, 7) = "Fortran"`

וכך הלאה.

ניתן ליצור מערכים עם עד 60 מימדים, והטבלה באה רק להמחיש את עניין המערכים הרב מימדיים. למשל אם יש לנו ערכי x, y, z של נקודה מסוימת, נוכל לייצג את הנקודה במערך בעל שלשה מימדים.

מערכים רב מימדיים דינמיים

ניתן להפוך מערך דינמי למערך רב מימדי, למשל:

`Dim Points() As Integer`

`ReDim Points(10, 10, 10) As Integer`

בשורה הראשונה, הצהרנו על מערך דינמי בשם Points, ובשורה השניה קבענו שהוא יהיה בן 3 מימדים שבכל אחד מהם 11 איברים (מ-0 עד 10).

כעת אם נרצה לשנות את מספר האיברים במערך הזה, נוכל לעשות זאת רק למימד האחרון, למשל:

`ReDim Preserve Points(10, 10, 15) As Integer`

כל ניסיון לשנות את מספר האיברים במימד אחד, חוץ מהאחרון יביא להודעת השגיאה: *Subscript out of range*.

הגבולות של המערך - הפונקציות LBound ו-UBound

לפעמים במהלך הפעלת התוכנית אנו לא יודעים מהו הגבול התחתון או העליון של המערך, מבחינת האינדקסים. לשם כך באות לעזרתנו שתי הפונקציות LBound ו-UBound.

הפונקציה UBound

פונקציה זו מחזירה את הערך של האינדקס הגבוה ביותר במערך שהועבר לה, למשל:

1	<code>Dim LargestIndex As Integer</code>
2	<code>Dim Points(5 To 10) As Integer</code>
3	<code>LargestIndex = UBound(Points)</code>

בשורה 1 הצהרנו על משתנה בשם LargestIndex מסוג מספר שלם (Integer).

בשורה 2 הצהרנו על מערך בשם Points, המכיל איברים מהאינדקס 5 עד האינדקס 10 (סך הכל 6 איברים).
 בשורה 3 בדקנו מהו האינדקס הגבוה ביותר במערך Points והצבנו אותו למשתנה LargestIndex.
 כעת ערך המשתנה LargestIndex יהיה 10, כי זהו האינדקס הגבוה ביותר במערך.
 אך אם יש לנו מערך רב מימדי, למשל:

```
Dim Points(5 To 10, 3 To 15, 2 To 20) As Integer
```

ואנו רוצים לדעת את ערך האינדקס הגבוה ביותר במימד השני, אז נעשה זאת כך:

```
LargestIndex = UBound(Points, 2)
```

וכעת המשתנה LargestIndex יקבל את הערך 15 (במספרי המימדים ויזואל בייסיק מתחילה לספור מ-1 ולא מ-0, ולכן מספרו של המימד השני הוא 2 ולא 1).

אם היינו רוצים למצוא את ערך האינדקס הגבוה ביותר במימד השלישי היינו עושים זאת כך:

```
LargestIndex = UBound(Points, 3)
```

הפונקציה LBound

פונקציה זו דומה מאוד לפונקציה UBound מלבד העובדה, שהיא מוצאת את ערך האינדקס הנמוך ביותר של המערך לדוגמא:

```
1 Dim SmallestIndex As Integer
```

```
2 Dim Points(5 To 10, 3 To 15, 2 To 20) As Integer
```

```
3 SmallestIndex = LBound(Points, 2)
```

ערך המשתנה SmallestIndex יהיה כעת 3, כי זהו האינדקס הקטן ביותר במימד השני במערך Points.
 ניתן כמובן להשתמש ב-LBound גם עם "מערכים שטוחים" (חד מימדיים), למשל:

```
Dim SmallestIndex As Integer
```

```
Dim Books(10 To 99) As String
```

```
SmallestIndex = LBound(Books)
```

ואז לא צריך לכתוב את מספר המימד שאת האיבר הקטן ביותר שלו אנו רוצים למצוא.

בדוגמא זו הערך שיקבל המשתנה SmallestIndex הוא 10, כי זהו האינדקס הנמוך ביותר במערך Books.

הצבת מערך למשתנה מסוג Variant – הפונקציה Array

ניתן במקרה הצורך להציב למשתנה מסוג Variant מערך, על ידי הפונקציה Array, ובכך לאחסן בו מערך, שיהיה ניתן לגשת לכל איבר שלו בעזרת אינדקס, כמו במערך רגיל.
 לדוגמא:

```
Dim Months As Variant
```

```
Months = Array("Jan", "Feb", "Mar", "Apr", "May", "Jun", "Jul", "Aug", "Sep", "Oct", "Nov", "Dec")
```

בדוגמא זו, הצהרנו בשורה הראשונה על משתנה בשם Months (חודשים) מסוג Variant.

בשורה השנייה הצבנו למשתנה Months את קיצורי שמות החודשים הלועזיים באמצעות הפונקציה Array. כעת אם נרצה לגשת לאיבר הרביעי שבמערך המאוחסן במשתנה Months נעשה זאת כך:

```
Dim MonthName As String
```

```
MonthName = Months(3)
```

כעת הצבנו למשתנה MonthName את האיבר הרביעי (כי מתחילים לספור מאפס, אלא אם כן השתמשנו ב-Option Base 1), ולכן ערכו של MonthName יהיה כעת "Apr".

מערך של אובייקטים

מערך של אובייקטים דומה למערך רגיל, רק במקום איברים שהם משתנים, האיברים שבו הם אובייקטים. מערך של אובייקטים יכול להיות מערך של כפתורי פקודה למשל, שלכולם יש את אותו השם, ומבדיל ביניהם רק האינדקס.

יצירת מערך אובייקטים

ישנן שתי דרכים עיקריות ליצור מערך של אובייקטים:


דרך ראשונה:

יצירת האובייקט בחלון העיצוב, העתקתו על ידי Copy ושכפולו על ידי Paste. במקרה זה ויזואל בייסיק תשאל אתכם, אם ברצונכם ליצור מערך של האובייקט הקיים, או ליצור אובייקט חדש. אם תבחרו ביצירת מערך, האובייקט המועתק יקבל אוטומטית את האינדקס 0 והאובייקט החדש את האינדקס 1.

בואו נתרגל את זה:

פתחו פרוייקט חדש בוויזואל בייסיק. צרו כפתור פקודה, כעת תסמנו את כפתור הפקודה (כך שיופיעו מסביבו הידיות הקטנות), ולחצו על המקשים Ctrl-1 או C ביחד, או תבחרו ב-Copy מהתפריט Edit. כעת לחצו על המקשים Ctrl-1 או V או בחרו ב-Paste מהתפריט Edit.

תיבת ההודעה הבאה תופיע:



בחרו ב-Yes ומערך האובייקטים ייוצר, כאשר האובייקט Command1 המקורי יקבל את האינדקס 0 והאיבר החדש יקבל את האינדקס 1.

כעת בדרך דומה, על ידי העתקה והדבקה, ניתן להוסיף אובייקטים נוספים למערך האובייקטים.

דרך שנייה:

לכו לתיבת התכונות, בחרו בכפתור הפקודה Command1, ושנו את תכונת האינדקס שלו ל-0, וכך בעצם יצרתם מערך אובייקטים בעל איבר אחד (וראשון) שהאינדקס שלו 0.

הוספת אובייקטים למערך אובייקטים בקוד

כדי להוסיף אובייקטים למערך אובייקטים (לאחר שכבר מערך האובייקטים נוצר באחת מהדרכים לעיל), יש להשתמש בפקודה Load למשל:

1	Load Command1(2)
2	Command1(2).Left = 0
3	Command1(2).Top = 0
4	Command1(2).Visible = True

בדוגמא זו הוספנו, בשורה 1, למערך האובייקטים של כפתורי הפקודה Command1, איבר חדש שהאינדקס שלו הוא 2.

בשורות 2-3, קבענו את מיקומו של כפתור הפקודה החדש במערך.

בשורה 4, קבענו כי כפתור הפקודה יהיה נראה (Visible = True), כי בדרך כלל איברים המוספים למערך אובייקטים, אינם נראים עד שלא הופכים אותם לנראים.

אם היינו מנסים להוסיף שנית את האיבר 2 למערך היינו מקבלים את הודעת השגיאה הבאה:

Object already loaded, כלומר שהאובייקט כבר נטען ולכן לא ניתן לטעון אותו שנית.

כדי להימנע ממצב זה, ניתן להטעין אובייקט תוך שימוש בפונקציה UBound, כך שתמיד האינדקס של האובייקט החדש, אכן יהיה חדש, למשל:

```
Load Command1(Command1.UBound + 1)
```

השימוש ב-UBound או ב-LBound במערך של אובייקטים שונה במעט ממערך רגיל, כפי שניתן לראות. במערך של אובייקטים בודקים את האינדקס הגבוה ביותר במערך, למשל כך:
Command1.Unbound

הסרת אובייקט ממערך של אובייקטים

כדי להסיר אובייקט ממערך אובייקטים יש להשתמש בפקודה UnLoad, למשל:

Unload Command1(0)

השורה למעלה, פשוט תמחק הן מהטופס והן מהזיכרון את האיבר 0 במערך האובייקטים Command1.

קוד לדוגמא למערך של אובייקטים ניתן להוריד מהאתר.

שאלה לחזרה

כיצד ניתן לדעת את האינדקס הגבוה ביותר במערך חד מימדי ("מערך שטוח")?

1. על ידי מדידת גובה המערך עם סרגל.
2. שואלים את המערך בנימוס מה האינדקס הגבוה ביותר.
3. על ידי הפונקציה UBound.
4. לא ניתן לדעת מהו האינדקס הגבוה ביותר במערך חד מימדי, אלא רק במערכים בעלי 289.5 מימדים בלבד.